

# SCIENCE & VIE

MENSUEL

N° 834 MARS 1987

ISSN 0036 8369

BELGIQUE 110 FB - ESPAGNE 475 Ptas - MAROC 17 Dh - TUNISIE 2,09 DT

SUISSE 5 FS - CANADA \$2,95 -

## 6 TESTS POUR DÉPISTER LE SIDA



**DARWIN CONTESTÉ**

M 2578 - 834 - 16,00 F



3792578016008 08340



1  
2  
3

JE GARDE 611 EN 1 COUP

VOUS : ●●●●

MOI : 611 ●●●●●●

# Jouons au "421"

INFORMATIQUE AMUSANTE

**N**ous vous offrons ici la possibilité d'apprendre à jouer au 421, sans devenir pour autant un pilier de bistrot. Car c'est chez vous, seul contre l'ordinateur, que vous pourrez vous faire la main. Mais rappelons, en bref, les règles de ce jeu de comptoir.

Le matériel est composé de 3 dés et 11 pions. Un premier joueur lance ses dés, une, deux ou trois fois au choix, de façon à obtenir la combinaison qui lui semblera la plus avantageuse par rapport au nombre de lancers. Le deuxième joueur lancera ensuite ses dés, obligatoirement le même nombre de fois que le premier, en essayant d'obtenir une meilleure combinaison. A chaque tour, c'est le joueur ayant perdu le tour précédent qui lancera les dés le premier et fixera ainsi le nombre de lancers.

Quand les deux, quatre ou six jets sont achevés, le perdant ramasse un ou plusieurs pions du gagnant, selon le score réalisé. Le but du jeu sera donc de réussir à faire prendre de ses pions par l'adversaire.

- La combinaison la plus forte est le "421", qui donne 7 pions à l'adversaire.

- Vient ensuite le "brelan", où les trois dés présentent un même chiffre. Le nombre de pions donnés est égal au nombre présenté par un dé ; il varie donc de 1 à 6.

- La combinaison suivante est "un chiffre plus deux as", où le nombre de pions donnés est égal à ce chiffre.

- Puis vient la "suite", où les chiffres présentés par les dés se suivent et le nombre de pions donnés est de 2.

• Enfin, nous trouvons les coups quelconques, où les trois dés forment trois chiffres différents qui, classés dans un ordre décroissant, donnent un nombre. Dans ce cas, c'est le joueur ayant réalisé le nombre le plus fort qui gagne et se débarrasse d'un pion.

En début de partie, tous les pions sont dans la "pioche". Ils sont distribués aux joueurs selon les résultats des jets jusqu'à ce que les 11 soient repartis entre les joueurs. Une fois ce stade atteint, les adversaires s'échangeront les

pions comme indiqué plus haut, jusqu'à ce que l'un d'entre eux les possède tous et perde ainsi la partie.

Hormis le premier lancer où le choix de l'ordre est laissé aux participants, en cours de partie c'est le joueur ayant réalisé le moins bon score qui relancera les dés. D'autre part, si à l'issue d'un jet les résultats sont identiques, les joueurs seront départagés sur un coup de dé unique. Le nombre de pions distribués sera alors fonction de la valeur de ce dernier jet.

En premier lieu nous définirons l'ensemble des symboles graphiques nécessaires à la représentation des dés. Au nombre de 20, il permettrons de les simuler en train de les rouler, et chacune de leurs faces. Les lignes 20 à 220 du programme sont réservées à cette opération. Ensuite la piste sera réalisée grâce à une fenêtre. Sa mise en place sera effectuée en ligne 230 et sa couleur en 240.

La présentation du reste de l'écran sera assurée par les lignes 250 à 350. L'ordinateur demandera

```

10 MODE 1:CLS
15 REM *****
16 REM * SYMBOLES GRAPHIQUES POUR LES DES. *
17 REM *****
20 SYMBOL AFTER 127
30 SYMBOL 128,3,13,17,97,129,145,129,129
40 SYMBOL 129,128,96,16,12,2,10,66,2
50 SYMBOL 130,134,152,160,201,96,16,12,3
60 SYMBOL 131,194,50,10,38,12,16,96,128
70 SYMBOL 132,1,2,4,8,24,36,66,137
80 SYMBOL 133,0,128,64,32,16,72,4,4
90 SYMBOL 134,128,128,64,32,18,8,5,2
100 SYMBOL 135,132,72,48,32,64,128,0,0
110 SYMBOL 136,0,0,63,32,32,32,32,32
120 SYMBOL 137,0,0,63,32,40,32,32,32
130 SYMBOL 138,0,0,63,32,40,32,32,33
140 SYMBOL 139,0,0,63,32,40,32,32,40
150 SYMBOL 140,0,0,248,8,8,8,8,8
160 SYMBOL 141,0,0,248,8,40,8,8,8
170 SYMBOL 142,0,0,248,8,40,8,8,40
180 SYMBOL 143,32,32,32,32,63,0,0,0
190 SYMBOL 144,32,32,40,32,63,0,0,0
200 SYMBOL 145,8,8,8,8,248,0,0,0
210 SYMBOL 146,8,8,40,8,248,0,0,0
220 SYMBOL 147,0,0,63,32,32,32,32,33
225 REM *****
226 REM * MISE EN PLACE DE LA FENETRE DE JEU *
227 REM *****
230 WINDOW #1,1,37,5,15
240 PAPER #1,2:CLS #1
245 REM *****
246 REM * PRESENTATION DE L'ECRAN *
247 REM *****
250 FOR I=1 TO 3
260 LOCATE 38,(I*3)+4:PRINT I
270 NEXT I
280 PEN 1
290 FOR I=15 TO 26
300 LOCATE 1,3:PRINT CHR$(231)
310 NEXT I
320 LOCATE 2,1:PRINT CHR$(128):CHR$(129):LOCATE 38,1:PRINT C
HR$(128):CHR$(129)
330 LOCATE 2,2:PRINT CHR$(130):CHR$(131):LOCATE 38,2:PRINT C
HR$(130):CHR$(131)
340 LOCATE 17,1:PRINT CHR$(137):CHR$(141):" ":CHR$(137):CHR$(
140):" ":CHR$(147):CHR$(140)
350 LOCATE 17,2:PRINT CHR$(144):CHR$(146):" ":CHR$(143):CHR$(
146):" ":CHR$(143):CHR$(145)
360 PEN #1,0:PAPER #1,2:LET DEB=0
365 REM *****
366 REM * CHOIX DU DEBUT DE PARTIE PAR LE JOUEUR *
367 REM *****
370 LOCATE 5,19:PRINT "VOULEZ-VOUS COMMENCER ? (o/n)"
380 LOCATE 1,22:PRINT "VOUS:";LOCATE 1,24:PRINT "MOI : "
390 LOCATE 15,22:PRINT " "
400 LOCATE 15,24:PRINT " "
410 LET FJ=0:LET FM=0:LET TM=0:LET TJ=0
420 LET K$=INKEY$
430 IF K$="" THEN GOTO 420
440 IF K$="o" THEN LET DEB=1
450 LOCATE 5,19:PRINT " "
460 LET CL=4:LET P=11:LET FJ=0:LET FM=0
470 IF FM=1 OR FJ=1 THEN GOTO 560
480 LET J=0:LET M=0
490 LOCATE 8,22:PRINT " ":LOCATE 8,24:PRINT " "
495 REM *****
496 REM * JEU *
497 REM *****
500 IF DEB=1 THEN LET J=1:GOSUB 1840
510 IF DEB=0 THEN LET M=1:GOSUB 1300
520 LET DEB=DEB+1:IF DEB=2 THEN DEB=0
530 IF J=1 AND M=1 AND LM<>LJ THEN LET CL=4:GOSUB 2060:GOTO
470
540 IF LM=LJ THEN LET CL=1:GOTO 470
545 REM *****
546 REM * REBOUCLAGE DE LA PARTIE EN COURS. *
547 REM *****
550 GOTO 500
560 IF FJ=1 THEN LOCATE 5,19:PRINT "BRAVO... VOUS AVEZ GAGNE
!"
565 REM *****
566 REM * CONTROLE DE FIN DE PARTIE. *
567 REM *****
570 IF FM=1 THEN LOCATE 5,19:PRINT "DESOLE, MAIS J'AI GAGNE!
"
580 LOCATE #1,10,3:PRINT #1,"POUR UNE AUTRE PARTIE"
590 LOCATE #1,10,6:PRINT #1,"TAPER UNE TOUCHE: MERCI. "
600 IF INKEY$="" THEN GOTO 600
605 REM *****
606 REM * DEMANDE DE NOUVELLE PARTIE. *
607 REM *****
610 GOTO 250
615 REM *****
616 REM * SOUS ROUTINE D'ANIMATION DU LANCE *
617 REM *****
620 IF GA=0 THEN LET A=INT(RND*6)+1
630 IF GB=0 THEN LET B=INT(RND*6)+1
640 IF GC=0 THEN LET C=INT(RND*6)+1
650 FOR I=35 TO 3 STEP -2
660 IF GA=1 THEN GOTO 690
665 SOUND 1,0,1,5,0,0,1/2
670 LOCATE #1,1,2:PRINT #1,CHR$(128):CHR$(129):" "
680 LOCATE #1,1,3:PRINT #1,CHR$(130):CHR$(131):" "
690 IF GB=1 THEN GOTO 720
695 SOUND 1,0,1,5,0,0,1/2
700 LOCATE #1,1,5:PRINT #1,CHR$(128):CHR$(129):" "
710 LOCATE #1,1,6:PRINT #1,CHR$(130):CHR$(131):" "
720 IF GC=1 THEN GOTO 750
725 SOUND 1,0,1,5,0,0,1/2
730 LOCATE #1,1,8:PRINT #1,CHR$(128):CHR$(129):" "
740 LOCATE #1,1,9:PRINT #1,CHR$(130):CHR$(131):" "
750 IF GA+GB+GC>1 THEN FOR T=1 TO 50:NEXT T
760 IF GA=1 THEN GOTO 790
765 SOUND 2,0,5,3,0,0,16
770 LOCATE #1,1,-1,2:PRINT #1,CHR$(132):CHR$(133):" "
780 LOCATE #1,1,-1,3:PRINT #1,CHR$(134):CHR$(135):" "
790 IF GB=1 THEN GOTO 820
800 LOCATE #1,1,-1,5:PRINT #1,CHR$(132):CHR$(133):" "
810 LOCATE #1,1,-1,6:PRINT #1,CHR$(134):CHR$(135):" "
820 IF GC=1 THEN GOTO 850
830 LOCATE #1,1,-1,8:PRINT #1,CHR$(132):CHR$(133):" "
840 LOCATE #1,1,-1,9:PRINT #1,CHR$(134):CHR$(135):" "
850 IF GA+GB+GC>1 THEN FOR T=1 TO 50:NEXT T
860 NEXT I
870 LET V=2
880 ON A GOSUB 940,960,980,1000,1020,1040
890 LET V=5
900 ON B GOSUB 940,960,980,1000,1020,1040
910 LET V=8
920 ON C GOSUB 940,960,980,1000,1020,1040
930 RETURN
935 REM *****
936 REM * REPRESENTATION DES DES APRES LE LANCE *
937 REM *****
940 LOCATE #1,2,V:PRINT #1,CHR$(147):CHR$(140)
950 LOCATE #1,2,V+1:PRINT #1,CHR$(143):CHR$(145):RETURN
960 LOCATE #1,2,V:PRINT #1,CHR$(137):CHR$(140)
970 LOCATE #1,2,V+1:PRINT #1,CHR$(143):CHR$(146):RETURN
980 LOCATE #1,2,V:PRINT #1,CHR$(138):CHR$(140)
990 LOCATE #1,2,V+1:PRINT #1,CHR$(143):CHR$(146):RETURN
1000 LOCATE #1,2,V:PRINT #1,CHR$(137):CHR$(141)
1010 LOCATE #1,2,V+1:PRINT #1,CHR$(144):CHR$(146):RETURN
1020 LOCATE #1,2,V:PRINT #1,CHR$(138):CHR$(141)
1030 LOCATE #1,2,V+1:PRINT #1,CHR$(144):CHR$(146):RETURN
1040 LOCATE #1,2,V:PRINT #1,CHR$(139):CHR$(142)
1050 LOCATE #1,2,V+1:PRINT #1,CHR$(144):CHR$(146):RETURN
1055 REM *****
1056 REM * SOUS D'EVALUATION DE LA VALEUR DU COUP *
1057 REM *****
1060 LET TR=0
1070 IF A=B AND B=C THEN LET L=111*A:GOTO 1200
1080 IF A=B AND A>C THEN LET L=(110*A)+C:GOTO 1200
1090 IF A=B AND A<C THEN LET L=(11*A)+(100*C):GOTO 1200
1100 IF A=C AND A>B THEN LET L=(110*A)+B:GOTO 1200
1110 IF A=C AND A<B THEN LET L=(11*A)+(100*B):GOTO 1200
1120 IF B=C AND B>A THEN LET L=(110*B)+A:GOTO 1200
1130 IF B=C AND B<A THEN LET L=(11*B)+(100*A):GOTO 1200
1140 IF A>B AND B>C THEN LET L=(100*A)+(10*B)+C:GOTO 1200
1150 IF A>C AND C>B THEN LET L=(100*A)+(10*C)+B:GOTO 1200
1160 IF B>A AND A>C THEN LET L=(100*B)+(10*A)+C:GOTO 1200
1170 IF B>C AND C>A THEN LET L=(100*B)+(10*C)+A:GOTO 1200
1180 IF C>A AND A>B THEN LET L=(100*C)+(10*A)+B:GOTO 1200
1190 IF C>B AND B>A THEN LET L=(100*C)+(10*B)+A:GOTO 1200
1200 IF DEB=1 THEN LOCATE 7,22:PRINT L
1210 IF DEB=0 THEN LOCATE 7,24:PRINT L
1220 PEN 2:LOCATE 10,17:PRINT " "
1230 IF L=221 THEN LOCATE 10,17:PRINT "'NENETTE'"
1240 IF L=654 OR L=543 OR L=432 OR L=321 THEN LET TR=1:LOCAT
E 10,17:PRINT "SUITE!"

```

alors si le joueur veut entamer la partie ou non (lignes 370 à 450). En fonction de la réponse fournie, DEB, variable permettant de gérer l'alternance des lancers, prendra pour valeur 1 ou 0, puis d'autres variables du jeu seront initialisées. CL correspond au nombre maximum de lancers autorisés pour un jet ; P au nombre de pions non encore distribués ; PJ à ceux du joueur ; PM à ceux de la machine, J permet de savoir si le joueur a lancé les dés et M si l'ordinateur les a lancés.

D'autres variables seront également utilisées tout au long du jeu. GA, GB et GC permettront de conserver des dés lors des second et troisième lancers d'un même jet. LM et LJ mémoriseront respectivement le score réalisé par la machine et par le joueur à l'issue d'un jet. La valeur TR sera fonction des figures obtenues : 0 pour un score sans particularité, 1 pour une "suite", 2 pour "un chiffre plus deux as", 3 pour un "brelan" et 4 pour le "421".

Cette valeur sera ensuite donnée

soit à TM si c'est l'ordinateur qui a réalisé le coup, soit à TJ si c'est le joueur. Enfin FJ et FM prendront respectivement la valeur 1 si le joueur ou la machine se sont débarrassés de l'ensemble de leurs pions.

Le jeu lui-même se déroule des lignes 500 à 600. En premier lieu, en fonction de la valeur prise par DEB, le programme exécutera soit la sous-routine de lancer du joueur, soit celle de l'ordinateur. Ensuite la valeur de DEB sera modifiée de manière à gérer l'alter-

```

1250 IF L=666 OR L=555 OR L=444 OR L=333 OR L=222 OR L=111 T
HEN LET TR=3:LOCATE 10,17:PRINT "BRELAN!"
1260 IF L=611 OR L=511 OR L=411 OR L=311 OR L=211 OR L=111 T
HEN LET TR=2
1270 IF L=421 THEN LET TR=4
1280 PEN 1
1290 RETURN
1295 REM *****
1296 REM * SOUS ROUTINE DE JEU DE L'ORDINATEUR *
1297 REM *****
1300 LET CP=0:LET GA=0:LET GB=0:LET GC=0
1310 LOCATE 5,19:PRINT "A MON TOUR DE JOUER !"
1320 GOSUB 620:GOSUB 1060
1330 LET GA=0:LET GB=0:LET GC=0
1340 LET CP=CP+1
1350 IF CP=4 OR CP=CL THEN GOSUB 1060:GOTO 1800
1360 IF TR>TJ THEN GOTO 1700
1370 IF CP=1 AND L<522 THEN GOTO 1410
1380 IF A=B THEN LET GA=1:LET GB=1:GOTO 1730
1390 IF A=C THEN LET GA=1:LET GC=1:GOTO 1730
1400 IF B=C THEN LET GB=1:LET GC=1:GOTO 1730
1410 IF TR>TJ AND L<633 THEN GOTO 1700
1420 IF A=1 AND B=1 THEN LET GA=1:LET GB=1:GOTO 1730
1430 IF A=1 AND C=1 THEN LET GA=1:LET GC=1:GOTO 1730
1440 IF B=1 AND C=1 THEN LET GB=1:LET GC=1:GOTO 1730
1450 IF A=4 AND B=2 THEN LET GA=1:LET GB=1:GOTO 1730
1460 IF A=4 AND B=1 THEN LET GA=1:LET GB=1:GOTO 1730
1470 IF A=2 AND B=4 THEN LET GA=1:LET GB=1:GOTO 1730
1480 IF A=2 AND B=1 THEN LET GA=1:LET GB=1:GOTO 1730
1490 IF A=1 AND B=4 THEN LET GA=1:LET GB=1:GOTO 1730
1500 IF A=1 AND B=2 THEN LET GA=1:LET GB=1:GOTO 1730
1510 IF A=4 AND C=2 THEN LET GA=1:LET GC=1:GOTO 1730
1520 IF A=4 AND C=1 THEN LET GA=1:LET GC=1:GOTO 1730
1530 IF A=2 AND C=4 THEN LET GA=1:LET GC=1:GOTO 1730
1540 IF A=2 AND C=1 THEN LET GA=1:LET GC=1:GOTO 1730
1550 IF A=1 AND C=4 THEN LET GA=1:LET GC=1:GOTO 1730
1560 IF A=1 AND C=2 THEN LET GA=1:LET GC=1:GOTO 1730
1570 IF B=4 AND C=1 THEN LET GB=1:LET GC=1:GOTO 1730
1580 IF B=4 AND C=2 THEN LET GB=1:LET GC=1:GOTO 1730
1590 IF B=2 AND C=4 THEN LET GB=1:LET GC=1:GOTO 1730
1600 IF B=2 AND C=1 THEN LET GB=1:LET GC=1:GOTO 1730
1610 IF B=1 AND C=4 THEN LET GB=1:LET GC=1:GOTO 1730
1620 IF B=1 AND C=2 THEN LET GB=1:LET GC=1:GOTO 1730
1630 IF A=B+1 OR B=A+1 THEN LET GA=1:LET GB=1:GOTO 1730
1640 IF A=C+1 OR C=A+1 THEN LET GA=1:LET GC=1:GOTO 1730
1650 IF B=C+1 OR C=B+1 THEN LET GB=1:LET GC=1:GOTO 1730
1660 IF A=1 THEN LET GA=1
1670 IF B=1 THEN LET GB=1
1680 IF C=1 THEN LET GC=1
1690 GOTO 1730
1700 IF CP<3 THEN LOCATE 5,19:PRINT "JE GARDE":L:"EN":CP:"C
OUP":CHR$(7)
1710 IF CP=2 THEN LOCATE 27,19:PRINT "S"
1720 GOTO 1800
1730 IF CP=3 THEN GOTO 1800
1740 FOR T=1 TO 1000:NEXT T
1750 IF GA=0 THEN LOCATE #1,2,2:PRINT #1," ":LOCATE #1,2,3:
PRINT #1," "
1760 IF GB=0 THEN LOCATE #1,2,5:PRINT #1," ":LOCATE #1,2,6:
PRINT #1," "
1770 IF GC=0 THEN LOCATE #1,2,8:PRINT #1," ":LOCATE #1,2,9:
PRINT #1," "
1780 IF CP=4 OR CP=CL THEN GOTO 1800
1790 GOTO 1320
1800 FOR T=1 TO 2000: NEXT T
1810 LET CL=CP
1820 LET TM=TR:LET LM=L
1830 RETURN
1835 REM *****
1836 REM * SOUS ROUTINE DU LANCE DU JOUEUR *
1837 REM *****
1840 LOCATE 5,19:PRINT "A VOUS, TAPÉZ UNE TOUCHE ."
1850 IF INKEY$="" THEN GOTO 1850
1860 CLS #1:LET CP=0
1870 LET GA=0:LET GB=0:LET GC=0
1880 GOSUB 620:GOSUB 1060
1890 LET CP=CP+1
1900 IF CP=3 OR CP=CL THEN GOTO 2010
1910 LET GA=1:LET GB=1:LET GC=1
1920 LOCATE 5,19:PRINT "RELANCER 1,2,3 ou 0 puis ' ' "
1930 LET K$=INKEY$
1940 IF K$="" THEN GOTO 1930
1950 IF K$="0" THEN GOTO 2010
1960 IF K$="1" THEN LET GA=0:LOCATE #1,2,2:PRINT #1," ":LOC
ATE #1,2,3:PRINT #1," "

```

```

1970 IF K$="2" THEN LET GB=0:LOCATE #1,2,5:PRINT #1," ":LOC
ATE #1,2,6:PRINT #1," "
1980 IF K$="3" THEN LET GC=0:LOCATE #1,2,8:PRINT #1," ":LOC
ATE #1,2,9:PRINT #1," "
1990 IF K$<>". " THEN GOTO 1930
2000 GOTO 1880
2010 LET CL=CP
2020 FOR T=1 TO 1000:NEXT T
2030 CLS #1
2040 LET TJ=TR:LET LJ=L
2050 RETURN
2060 IF P=0 THEN GOTO 2310
2065 REM *****
2066 REM * ROUTINE DE DISTRIBUTION DES PIONS *
2067 REM *****
2070 IF TM>=TJ THEN GOTO 2180
2080 LET DEB=0
2090 IF TJ=0 THEN LET PM=PM+1:LET P=P-1:GOTO 2450
2100 IF TJ=1 AND P>1 THEN LET PM=PM+2:LET P=P-2:GOTO 2450
2110 IF TJ=1 THEN LET PM=PM+P:LET P=0:GOTO 2450
2120 IF TJ=2 AND P>INT(LJ/100) THEN LET PM=PM+INT(LJ/100):LE
T P=P-INT(LJ/100):GOTO 2450
2130 IF TJ=2 THEN LET PM=PM+P:LET P=0:GOTO 2450
2140 IF TJ=3 AND P>INT(LJ/100) THEN LET PM=PM+INT(LJ/100):LE
T P=P-INT(LJ/100):GOTO 2450
2150 IF TJ=3 THEN LET PM=PM+P:LET P=0:GOTO 2450
2160 IF TJ=4 AND P>7 THEN LET PM=PM+7:LET P=P-7:GOTO 2450
2170 IF TJ=4 THEN LET PM=PM+P:LET P=0:GOTO 2450
2180 IF TM=TJ THEN GOTO 2290
2190 LET DEB=1
2200 IF TM=0 THEN LET PJ=PJ+1:LET P=P-1:GOTO 2450
2210 IF TM=1 AND P>1 THEN LET PJ=PJ+2:LET P=P-2:GOTO 2450
2220 IF TM=1 THEN LET PJ=PJ+P:LET P=0:GOTO 2450
2230 IF TM=2 AND P>INT(LM/100) THEN LET PJ=PJ+INT(LM/100):LE
T P=P-INT(LM/100):GOTO 2450
2240 IF TM=2 THEN LET PJ=PJ+P:LET P=0:GOTO 2450
2250 IF TM=3 AND P>INT(LM/100) THEN LET PJ=PJ+INT(LM/100):LE
T P=P-INT(LM/100):GOTO 2450
2260 IF TM=3 THEN LET PJ=PJ+P:LET P=0:GOTO 2450
2270 IF TM=4 AND P>7 THEN LET PJ=PJ+7:LET P=P-7:GOTO 2450
2280 IF TM=4 THEN LET PJ=PJ+P:LET P=0:GOTO 2450
2290 IF LJ>LM THEN GOTO 2080
2300 IF LJ<LM THEN GOTO 2190
2305 IF LJ=LM THEN LET CL=2:GOTO 500
2310 IF TM>=TJ THEN GOTO 2370
2315 LET DEB=0
2320 IF TJ=0 THEN LET PM=PM+1:LET PJ=PJ-1:GOTO 2450
2330 IF TJ=1 THEN LET PM=PM+2:LET PJ=PJ-2:GOTO 2450
2340 IF TJ=2 THEN LET PM=PM+INT(LJ/100):LET PJ=PJ-INT(LJ/100
):GOTO 2450
2350 IF TJ=3 THEN LET PM=PM+INT(LJ/100):LET PJ=PJ-INT(LJ/100
):GOTO 2450
2360 IF TJ=4 THEN LET PM=PM+7:LET PJ=PJ-7:GOTO 2450
2370 IF TM=TJ THEN GOTO 2430
2375 LET DEB=1
2380 IF TM=0 THEN LET PJ=PJ+1:LET PM=PM-1:GOTO 2450
2390 IF TM=1 THEN LET PJ=PJ+2:LET PM=PM-2:GOTO 2450
2400 IF TM=2 THEN LET PJ=PJ+INT(LM/100):LET PM=PM-INT(LM/100
):GOTO 2450
2410 IF TM=3 THEN LET PJ=PJ+INT(LM/100):LET PM=PM-INT(LM/100
):GOTO 2450
2420 IF TM=4 THEN LET PJ=PJ+7:LET PM=PM-7:GOTO 2450
2430 IF LJ>LM THEN GOTO 2315
2440 IF LM>LJ THEN GOTO 2375
2445 IF LM<LJ THEN LET CL=2:GOTO 500
2450 IF PJ<=0 AND P=0 THEN LET FJ=1:RETURN
2460 IF PM<=0 AND P=0 THEN LET FM=1:RETURN
2470 LOCATE 15,22:PRINT " "
2480 LOCATE 15,3:PRINT " "
2490 LOCATE 15,24:PRINT " "
2500 FOR I=15 TO 14+P
2510 LOCATE I,3:PRINT CHR$(231)
2520 NEXT I
2530 FOR I=15 TO 14+PJ
2540 LOCATE I,22:PRINT CHR$(231)
2550 NEXT I
2560 FOR I=15 TO 14+PM
2570 LOCATE I,24:PRINT CHR$(231)
2580 NEXT I
2590 PRINT CHR$(7)
2600 FOR T=1 TO 1000:NEXT T
2610 LET TJ=0:CLS #1
2620 RETURN

```

nance des lancers, puis le programme vérifiera que les deux participants ont bien lancé les dés. Enfin il sera vérifié que les résultats sont différents. Si l'un des participants à épuisé son stock de pions, le programme l'indiquera et demandera si une autre partie est souhaitée. En cas contraire le jeu rebouclera.

Voyons, à présent, les six sous-routines utilisées.

**Le lancer des dés** occupe les lignes 620 à 930 du programme. En premier lieu, les valeurs que prendront les dés seront tirées aléatoirement par les lignes 620, 630 et 640. Viendra alors l'animation chargée de figurer les dés en train de rouler. Notons que les valeurs prises par GA, GB et GC seront prises en compte afin de ne pas relancer un dé devant être conservé. Enfin elle fera appel au groupe de sous-routines figurant les dés arrêtés.

**La représentation des dés après le lancer** est possible par un groupe de sous-routines. Chacune d'entre elles a pour mission de présenter une des faces d'un dé. Nous trouverons donc six cas différents puisqu'un dé possède six faces. Le choix du cas employé sera effectué par les lignes 880, 900 et 920 de la sous-routine du lancer des dés.

**L'évaluation de la valeur du coup.** Cette sous-routine, placée des lignes 1060 à 1290, a pour mission de remettre en ordre décroissant les chiffres déterminés par le lancer des dés et de détecter s'il s'agit d'un score ordinaire, d'une suite, etc. A l'issue de cette sous-routine, L prendra pour valeur le nombre obtenu en classant par ordre décroissant les chiffres portés par les dés et TR indiquera le type de résultat issu du jet.

**Le jeu de l'ordinateur.** Il s'agit de la sous-routine de décision de la machine. Placée des lignes 1300 à 1830, elle prendra en compte le nombre de coups autorisés (CL) ainsi que le score obtenu par le joueur si ce dernier a lancé les dés auparavant. A l'issue de cette sous-routine les valeurs LM et TM seront fonction des résultats obtenus. De même, si c'est l'ordinateur qui a lancé le premier, la valeur de CL sera fonction du nombre de jets choisis par la machine.

**Le lancer du joueur,** dans cette

sous-routine, est placé des lignes 1840 à 2060 ; nous ferons régulièrement appel à la variable K\$. Celle-ci a pour fonction d'indiquer au programme les touches frappées au clavier. De là, les divers choix du joueur pourront être pris en compte. Ici encore, CL sera utilisée pour borner le nombre de lancers possibles et, à l'issue de cette sous-routine, LJ et TJ prendront des valeurs en fonction du score obtenu.

**La distribution des pions** comporte deux parties. La première, placée des lignes 2070 à 2305, assure la répartition des pions aux deux participants tant que les 11 pions de début de jeu n'ont pas été distribués. La seconde, placée des lignes 2310 à 2460, assure l'échange des pions entre les joueurs durant le reste du jeu.

Enfin, trois boucles FOR... NEXT associées à des instructions PRINT permettront de contrôler l'affichage de la répartition des pions. A l'issue de cette sous-routine, les valeurs de PJ et PM seront remises à jour. De même, si l'un des participants a épuisé son stock de pions ce fait sera signalé en donnant à FJ, ou FM, la valeur 1.

Une fois le programme entièrement tapé, RUN sera demandé. Après avoir affiché sur l'écran l'ensemble des éléments du jeu l'ordinateur indiquera : "Voulez-vous commencer ? (o/n)". La réponse lui sera fournie en frappant "o" pour oui ou "n" pour non. Attention, le mode "cap lock" ne devra pas être utilisé ici. Ensuite, soit l'ordinateur lancera les dés, soit il indiquera "Tapez une touche." Cette opération effectuée, vos dés seront lancés une première fois. L'ordinateur indiquera alors : "RELANCER" 1, 2, 3 ou 0 puis '.'. On indiquera donc le numéro des dés à relancer en le tapant au clavier. Les dés concernés s'effaceront automatiquement de la piste dès que leur numéro aura été demandé.

Une fois ce choix terminé il suffira de taper "." (point) pour que le nouveau lancer soit effectué. Notons que si tous les dés doivent être conservés il suffira de taper "0". L'alternance des coups ainsi que la distribution des pions sont contrôlées par l'ordinateur. Aucune intervention ne sera donc nécessaire pour assurer ces fonctions.

**Henri-Pierre Penel**